
Abstract

This report describes a Larch interface language (LM3) for the Modula-3
programming language. LM3 is a complete example of a Larch interface
language and addresses areas previously ignored in interface language de�-
nition, such as the speci�cation of non-atomic procedures and object types.

We give a complete de�nition of the syntax and illustrate it with some
straightforward examples. We also give translation functions from LM3
speci�cations to Larch Shared Language traits and show their use for type
checking. Finally, we present example speci�cations of standard Modula-3
interfaces.

To remove the possibility of misunderstanding, this report presents LM3

using its base syntax and does not use any syntactic sugar. In practice, such

sugar is convenient and the checker accepts a sugared form as well as the

raw form presented here.

Chapter 1

Introduction

1.1 Background

Larch provides a family of speci�cation languages that may be used to
specify program interfaces, plus a collection of tools to aid in constructing
correct speci�cations. Larch speci�cations are given in two parts:

1. the Shared Language tier, which uses an algebraic speci�cation lan-
guage to describe the properties of the basic data types and operators
used in the speci�cation. The Larch Shared Language (LSL)[6] is
common to all Larch speci�cations.

2. the interface language tier, which is speci�cally related to the program-
ming language being used. An interface language contains constructs
that are appropriate to the programming language and uses the traits
speci�ed in the LSL tier to describe the properties of the interface.

This report describes an interface language (LM3) that is designed for use
with the Modula-3 language[4]. It is assumed that the reader is already
familiar with Modula-3, LSL, and the general ideas of interface speci�cation
(as in, say, [13]).

Previous publications have documented Larch interface languages, for
example [12]. LM3 follows the general style of this previous work, but
addresses several additional features that are becoming common in new
programming languages.

� Modula-3 allows higher order procedures (i.e., those which take pro-
cedure parameters or return procedure results). Since procedures are

represented by their speci�cations rather than by their values, we need
a way of associating speci�cations with such parameters.

� Modula-3 allows object type hierarchies. This means LM3 has to
inherit speci�cations from supertypes.

� Modula-3 supports concurrently executing threads of control, there-
fore, LM3 has to be able to specify non-atomic procedures.

LM3 addresses each of these areas. Where possible, the extra features
follow the style and philosophy of previous Larch work.

1.2 The relationship betweenModula-3 and LM3

A Modula-3 module that provides an externally usable interface usually
consists of two �les:

1. an interface �le, with the extension .i3, which de�nes the exported
de�nitions of the module;

2. an implementation �le, with the extension .m3, which gives the full
code of the module.

LM3 speci�cations are placed in the interface �le. Clients of the module
see only the .i3 �le. All of the information necessary for understanding
and using the module should be presented in this �le. Without LM3,
the interface is usually supplemented by comments describing the intended
action of the procedures, and so forth. From the point of view of the clients,
the LM3 speci�cation provides a more precise description of the functionality
of the interface and can replace some of the detailed textual comments,
although general comments should still be used to supplement the formal
text. From the point of view of the programmers of the .m3 �le, the LM3
speci�cation provides a contract to which they must implement.

LM3 annotations decorate standard Modula-3 as pragmas. The main
intention of pragmas, according to the Modula-3 report, is to provide `hints
to the implementation; they do not a�ect the language semantics.' Since
unrecognized pragmas are ignored by the compiler, they provide a convenient
way of attaching the speci�cation information.

We believe that speci�cations should be kept with the programs they
are intended to describe. LM3 interface speci�cations are legal Modula-3
interfaces. By making the speci�cation an integral part of the program,

we intend to remind the programmer of its existence at all stages in the
life of the interface. Previous Larch interface languages have placed the
speci�cations within comments but doing this has tended to de-emphasize
their importance in the minds of some. We expect pragmas to be taken
more seriously.

Chapter 2 introduces the constructs of LM3 and illustrates their use
with simple examples. The complete syntax of LM3 is given in Chapter 3.
Chapter 4 de�nes the LSL traits used by LM3 and gives the translation
functions for LM3 constructs. Chapter 5 gives examples of complete LM3
interface speci�cations.

Chapter 2

The LM3 speci�cation

language

An LM3 interface speci�cation consists of a Modula-3 interface de�nition
together with some annotations within pragma brackets. A speci�cation
contains type speci�cations, variable and constant declarations, procedure
speci�cations and an interface invariant. In the following sections, we con-
sider these in turn.

We introduce the following grammatical conventions:

� terminal symbols | term

� LM3 and Modula-3 keywords | KEYWORD

� non-terminals | nonTerm

� foo,� means zero or more foo's separated by a ,

� foo;+ means one or more foo's separated by a ;

� [foo] means zero or one foo

� extensions to the Modula-3 grammar are indicated as foo

2.1 Interfaces

The top-level unit in an LM3 speci�cation, like that of a Modula-3 program,
is the interface. A Modula-3 interface is annotated with a list of traits

that de�ne the meaning of the symbols used within predicates to specify
the functionality of the interface components. Any renaming necessary to
remove ambiguity must be done here, using LSL mechanisms.

An interface speci�cation has the form:

interface ::= INTERFACE ident ; [traitUse] imports�

[intConstraints] declaration� END ident .

traitUse ::= <� USING traitRef,+ �>

imports ::= [FROM ident] IMPORT ident,+ ;

intConstraints ::= <� initial �> j <� invar �> j <� initial invar �>

declaration ::= constDecl j varDecl j typeDecl j exceptionDecl
j procDecl j privateVarDecl j : : :

initial ::= INITIALLY lm3Predicate

invar ::= INVARIANT lm3Predicate

There may be an initial condition and an invariant associated with
the interface. In an interface that declares variables, the initial condition
constrains the initial value of the variables. The invariant would normally
be used to state relationships that must always be maintained between these
variables. It may also be used to specify relationships between procedures
within the interface.

Since grammar fragments are not the most enlightening way of under-
standing a language, we develop a simple example1 as we introduce each
new construct. A skeleton of the interface is given as:

INTERFACE Stack;

<* USING Stack(Real for E, RealStack for C) *>

(* declarations of exported type and procedures - see below *)

END Stack.

The USING clause associates all symbols used in the speci�cation with
those found in the trait named Stack, with the appropriate renaming. All
interfaces implicitly use a trait LM3Trait which gives the de�nitions of the
primitive operations of Modula-3. See Chapter 4. The trait Stack, from the
Larch Shared Language Handbook[8], is given in Appendix A.

1Our excuse for using the ubiquitous stack example is that we are following the
Modula-3 Report, which uses Stack as its example of an interface.

2.2 Declarations

Declarations in Modula-3 interfaces include constants, types, variables, ex-
ceptions and procedures. In this section, we describe the LM3 speci�cations
for each since the form of the speci�cation depends on the kind of declara-
tion.

2.2.1 Constants

An interface may export any number of constants. The grammar for con-
stant declarations is:

constDecl ::= ident [: type] = constExpr

A constant declaration for LM3 is just a Modula-3 constant declaration,
with the restriction that the constExprmust be a term of the LM3 expression
language.

For example, if we wished to give an upper bound to the stack, we could
write:

CONST MaxSize = 100;

2.2.2 Variables

LM3 adds no extra information to the declaration of exported variables.
Any restrictions may be placed in either the type or the interface invariant.
The grammar for a variable declaration is:

varDecl ::= VAR vd+

vd ::= ident,+ : type [initialVal] ;

initialVal ::= := expr

Exporting variables is not common in Modula-3 interfaces.

2.2.3 Private variables

It is often the case that the speci�cation uses information that does not have
to be accessible to the implementation.

To facilitate this, LM3 allows the declaration of private variables, which
are notionally part of the state but which may be referenced only within

speci�cations. These variables exist only within the speci�cation domain
and are associated with LSL sorts, not with types. For convenience, the
obvious sorts are available to represent the common programming language
types.

The grammar for private variables is:

privateVarDecl ::= <� PRIVATE ( ident,+ : sort; varSpec)+ �>

varSpec ::= [initial] [invar]

2.2.4 Types

Modula-3 has a rich space of types, including a notion of subtyping. All of
the base types and the type constructors of the language are associated with
LSL sorts in LM3Trait.

A type declaration may be annotated with a number of things. Not all
will be appropriate in all cases. A fully annotated type declaration has the
following form:

typeDecl ::= TYPE td+

td ::= ident [typeSpec] subtypeReln type ;

typeSpec ::= <� BASED ON [ident:]sort [initial] [invar] �>

subtypeReln ::= = j <:

type ::= ident j arrayType j recordType j : : :

The �rst clause associates the type with a sort, which must be de�ned
in one of the traits in the USING list. In particular, for any variable v of type
T which is based on sort S, the value of v must be equal to a term of sort
S for which T's invariant is true.

The initial clause, indicated by INITIALLY, introduces a predicate that
must hold for initial values of a type. For a simple type this predicate
constrains the initial values supplied in variable declarations. For an object
type it is a constraint on the result of calls to NEW. Satisfaction of this
constraint is an obligation of clients that use the type.

The invar clause, indicated by INVARIANT, introduces a predicate that is
notionally conjoined to the pre- and post-condition of all procedures that

may reference an element of the type. Such a procedure may assume the
invariant on invocation and must guarantee it on exit. Object types, where
invariants are inherited, are discussed separately in Section 2.3.3.

The ident represents a variable bound by universal quanti�cation over
the type.

Each type declared in the interface is associated with a sort. There are
two categories of types:

1. types with modi�able values (REF types). That is T <: REFANY or ROOT.
If such a T is BASED ON a sort S, the sort actually used to represent the
type is SRef.

2. types with unmodi�able values (VAL types). In this case, there is no
indirection and the sort is the one given in the BASED ON. Parameters
of such a type are unmodi�able (and it is a checked error to try to
modify a parameter of a VAL type). Modi�cations of such parameters
are permitted only if they are declared as VAR. In this case the sort of
the parameter is SVar which can be regarded as implicitly introducing
an extra level of reference.

If we expand the Stack example to include the declaration of the type,
we see:

INTERFACE Stack;

<* USING Stack(Real for E, RealStack for C) *>

TYPE T <* BASED ON RealStack *>

<: REFANY;

(* declarations of exported procedures - see below *)

END Stack.

For the example above, any variable of type Stack has sort RealStackRef.
To get the value of such a variable, which has sort RealStack (which is pro-
duced by performing the given renaming on the trait Stack), we dereference
the variable in the appropriate state using either __\pre or __\post.

2.2.5 Procedures

Modula-3 allows the use of a variety of constructs within procedures, in-
cluding exceptions and threads. Therefore, procedure declarations may

10



be annotated with a number of predicates. The meaning of a procedure
speci�cation is given as a predicate on a sequence of state pairs2.

Atomic procedures

Most procedures written in Modula-3 are atomic. Since the speci�cation of
an atomic procedure is signi�cantly simpler than in the non-atomic case, we
consider this �rst.

If there are no exceptions and the procedure is atomic and unsynchro-
nized, the grammar is:

procDecl ::= PROCEDURE ident ([signature]) [: type] ;
[<� procSpec �>]

signature ::= f[paramType] ident,+ : type [initialVal] ;g+

paramType ::= VALUE j VAR j READONLY

procSpec ::= [globals] [privates] [letDecl]

[prePred] [modi�es] postPred

globals ::= ((WR j RD) ident : type;)+

privates ::= PRIVATE varDecl+

letDecl ::= LET let IN

let ::= ident BE term,+

prePred ::= REQUIRES lm3Predicate

modi�es ::= MODIFIES term,+

postPred ::= atomicPostPred

atomicPostPred ::= ENSURES lm3Predicate

The components of the procedure speci�cation are:

globals Declarations of all of the variables in the global state that are ref-
erenced by this procedure. For most purposes, this can be considered

2For a sequential language, this would be a relation on a single state pair. The
introduction of concurrency means that we need to extend to a more complex semantic
domain, where for each atomic action, a pair represents a state at the start of the action
(pre) and one at the completion (post).

as an implicit extension to the parameter list. Global variables are
annotated to indicate their intended use: globals may be indicated to
be WR (writable) if they may be modi�ed or RD (read-only) if it is not
intended to change the value. It is an error to modify a global variable
that has been declared to be RD.

privates Private variables local to each invocation.

letDecl Local shorthands de�ned by use of the LET construct. This is purely
a syntactic substitution mechanism.

prePred a REQUIRES clause that de�nes the precondition of the procedure.
This is a predicate that the caller must ensure is true in the state from
which the procedure is invoked. If it is not , then nothing is guaranteed
about the result, including termination. If this clause is omitted then
it defaults to true, implying the procedure may be invoked from any
state. The REQUIRES clause may reference only variables in the pre

state.

modi�es a MODIFIES clause that identi�es the state components that the
procedure is allowed to modify. If there is no MODIFIES clause, then
the procedure may not modify anything. If the procedure is allowed
to modify anything to which it has access, the shorthand is MODIFIES

ALL. Modi�cation must be consistent with the type of the parameter.
For example, it is an error to mention a VAL or READONLY parameter in
the MODIFIES list, remembering the default for M3 parameters is VAL.
Only global variables declared as WRmay be mentioned in the MODIFIES
clause. This clause is a list of terms whose values are elements of a
type that is modi�able (normally, a REF). From this list, a predicate
asserting the validity of such modi�cations is derived.

postPred an ENSURES clause that gives the postcondition of the procedure.
This is a predicate over the pair of states (the state on invocation and
the state on termination) that must be true on exit from the speci�ed
procedure. Variables in the ENSURES clause may refer to values in
both states and so are quali�ed with either __\pre or __\post. The
unnamed return value of a procedure is represented by the pseudo
variable RESULT which only has meaning in the �nal state and so must
be quali�ed with __\post.

Formally, the meaning of such a speci�cation of a procedure, when fully
expanded, is given by the predicate :

prePred) (modPred^ postPred)

As an example, we add some functionality to our evolving Stack. Since
we are following the example that can be found in the Modula-3 report, in
which all procedures return a new Stack rather than modify the existing
one, the speci�cation is:

PROCEDURE Pop(VAR s:T): REAL;

<* REQUIRES NOT(isEmpty(s\pre\pre))

MODIFIES s

ENSURES s\post\post = pop(s\pre\pre)

AND RESULT\post = top(s\pre\pre) *>

This tells us that this procedure should be called only when the value of
s is not the empty stack, that the value of s may be modi�ed and that the
result and the �nal value of s will be the top and pop (from the Stack trait)
of the initial value of s, respectively.

Since s is declared to be a VAR parameter, that implicitly adds a level of
indirection. In other words, s is associated with the sort SRef rather than
S. This explains the two uses of __\pre to get to a stack value: the �rst to
dereference the VAR giving a SRef; the second to dereference this giving a S
(remembering that T <: REFANY).

An aside

This speci�cation of Stack is somewhat unusual. Since this has caused some
confusion amongst some readers of this report, we'll discuss it a little.

More typically, one might expect the procedure to modify the existing
Stack, rather than deliver a new one. To understand the contrast between
these two styles, the speci�cation of a Pop1 function in which the Stack is
not a VAR parameter is:

PROCEDURE Pop1(s:T): REAL;

<* REQUIRES NOT(isEmpty(s\pre))

MODIFIES s

ENSURES s\post = pop(s\pre)

AND RESULT\post = top(s\pre) *>

Here, the given Stack is changed. This is allowed since Stack is a REF

type. We follow the former example in the rest of this report, since the extra
level of indirection forces greater care in the speci�cation and so serves our
pedagogical purpose better, but feel we should point out that the strangeness
is due to the desired behavior of the example rather than to the speci�cation
language.

Keyword predicates

Since the following example makes use of them, this is an appropriate place
to mention LM3's keyword predicates. These keywords form an important
part of the terms used to build predicates. Keyword predicates are :

UNCHANGED(v1 : : : vn), which asserts that the �nal values of these variables
are equal to their initial values.

FRESH(foo), meaning that the storage assigned to foo is not shared with
anything else in the state.

CHECKEDRTE, which is semantically equivalent to true but it warns the
implementer that a checked run time error would occur

An interlude: the Stack

By this point, we have su�cient mechanism to complete our Stack example.
Fitting together the pieces seen previously and adding the obvious proce-
dures we get:

INTERFACE Stack;

<* USING Stack (REAL for E, RealStack for C) *>

TYPE T <* BASED ON RealStack *>

<: REFANY;

PROCEDURE Pop(VAR s:T): REAL;

<* REQUIRES NOT(isEmpty(s\pre\pre))

MODIFIES s

ENSURES s\post\post = pop(s\pre\pre)

AND RESULT\post = top(s\pre\pre)

*>

PROCEDURE Push(VAR s: T; x: REAL);

<* MODIFIES s

ENSURES s\post\post = push(s\pre\pre, x)

*>

PROCEDURE Create(): T;

<* ENSURES isEmpty(RESULT\post) AND FRESH(RESULT) *>

END Stack.

This gives all the functional information that a client of this interface
should ever need to know. In practice, the speci�cation should be supple-
mented by textual comments, telling clients the (equally important) non-
functional things they need to know.

Procedures with Exceptions

One of the common programming techniques in Modula-3 is the use of
exceptions for abnormal termination of a procedure. The programming
language allows you to declare the set of exceptions that may be raised
by a procedure. The speci�cation language permits you to describe an
alternative result by using an EXCEPT clause, with guarded predicates that
may be satis�ed in place of the normal post condition. If any guards are
true, then the procedure must satisfy the exception predicate of one of them,
rather than the normal ENSURES predicate.

There is also the possibility of an exception being raised by a lower
level of the program. The circumstances that lead to such an \abstraction
failure" exception generally cannot be speci�ed in terms of the state that is
accessible to the caller. As far as the caller is concerned, these exceptions
may be raised \arbitrarily". They are speci�ed by UNLESS clauses. If a

procedure speci�cation has such clauses, then its �nal value may be either
the value of one of the UNLESS predicates or that of the ENSURES/EXCEPT

clause.
A state variable, RAISE, represents the value of a raised exception in

the post state. This may take the value of any legitimate exception or the
special value RETURN which indicates normal termination of the procedure.
As a syntactic convenience, when an EXCEPT or an UNLESS clause is present,
the term `RAISE = RETURN' is implicitly conjoined to the ENSURES clause.

The following extensions are made to the procDecl grammar:

procDecl ::= : : : [raisesList] ; [procSpec]

raisesList ::= RAISES f ident,� g

atomicPostPred ::= ENSURES lm3Predicate [except] [unless]

except ::= EXCEPT fguardPredicate ) exceptionPredicateg |+

unless ::= UNLESS exceptionPredicate |+

The keywords hopefully imply the correct interpretation of the postcon-
dition, which is:

� the post predicate is true on exit

� except if any of the guards are true, then the corresponding exception
predicate is true on exit. If more than one guard is true, a non-
deterministic choice is allowed.

� unless one of the unguarded exception predicates is true on exit.

For example, for a procedure (loosely) speci�ed as:

PROCEDURE x( ... );

<* REQUIRES prePred

MODIFIES modPred

ENSURES postPred

EXCEPT g1 => ex1 | g2 => ex2

UNLESS ex3 | ex4 *>

the meaning is :

prePred) (modPred ^

(:(g1_ g2)^ postPred ^ RAISE = RETURN)

_(g1 ^ ex1)
_(g2 ^ ex2)
_ ex3 _ ex4)

If, for example, our stack raised an exception on trying to push an object
into a stack whose size was MaxSize, the procedure could be speci�ed as:

EXCEPTION StackOverflow;

PROCEDURE Push(VAR s: T; x: REAL) RAISES{StackOverflow};

<* MODIFIES s

ENSURES s\post\post = push(s\pre\pre, x)

EXCEPT size(s\pre\pre) = MaxSize =>

RAISE\post = StackOverflow

AND UNCHANGED(s)

*>

Non-atomic procedures

The speci�cation of non-atomic procedures is less well understood than that
of atomic procedures. This area is a large part of our on-going research, and
while the mechanisms proposed in this section are suitable for our current
needs, it is likely that in the future, as the technology matures, we will adopt
an approach based more closely on Lamport's Temporal Logic of Actions[9].

Post-conditions of atomic procedures range over exactly two states, a pre
state and a post state. However, since Modula-3 allows concurrent threads
of activity within an address space, this model is not su�ciently general
for describing all Modula-3 procedures. In particular, we need to be able to
describe intermediate states, which may be visible to other threads. To allow
this, we specify a non-atomic procedure as being composed of a number of
separate atomic actions. Each action is modeled as a relation on a pair of
states, as before. The entire procedure can now be speci�ed in terms of the
sequence of state pairs, each pair representing one atomic action.

Some non-atomic procedures require a means of referring to the currently
executing thread. This is designated by the keyword, CURRENT.

WHEN Predicate

Concurrency adds the need to specify when an action may take place, as
well as what it does. This is given by a WHEN predicate. If a WHEN clause is
given, the action is allowed only when the predicate is true in its pre state,

which since there is concurrent activity may not be the same as the state at
the call. The grammar is extended to include:

when ::= WHEN lm3Predicate

as an optional component of a procedure speci�cation.

Composite procedures

Explicit composition The simplest extension is to allow procedures that
can be described as an explicit composition of subsidiary atomic actions. The
grammar is extended to:

postPred ::= atomicPostPred j compositePostPred

compositePostPred ::= COMPOSITION OF ident;+ END action+

action ::= ACTION ident [when] atomicPostPred

Examples of the use of composition can be found in Section 5.1.

Arbitrary composition

In this case, we describe the actions in much the same way, except that we
do not know how many actions take place. We introduce a further notion of
de�ning an action that may take place an arbitrary number of times3. Such
an action is followed by the symbol �. As a syntactic convenience, we allow
A; A* to be written as A+.

The full grammar for non-atomic procedures is therefore:

postPred ::= atomicPostPred j compositePostPred

compositePostPred ::= COMPOSITION OF acts ;+ action+ END

acts ::= ident[� j +] ; j ( acts )

3We recognize that this extension is still not fully general. However, we appeal to
a remark that Jim Horning attributes to Leslie Lamport, paraphrased as \90% of all
procedures should appear atomic to their clients". We claim that the mechanism proposed
here gives another 8% and we are cheerfully ignoring the remaining few for now.

This allows certain actions, say the �rst or the last, to be speci�cally
constrained, while all other actions may be speci�ed using the same predi-
cate.

For example, the speci�cation fragment:

COMPOSITION OF A+; B; C*; D END

tells us that this procedure can be modeled as: at least one A action, a B
action, maybe some C actions and �nally a D action.

2.3 Other Modula-3 features

Modula-3 is a modern language with some advanced features that require
special attention in the speci�cation language. Some, such as concurrent
threads, have already been addressed above. The rest are gathered together
into the following sections.

2.3.1 Procedure parameters

Modula-3 allows the programmer to pass procedures as parameters to other
procedures. This feature is well understood at the program level since one
can just call the passed procedure just like any other procedure. However, at
the speci�cation level, it is the procedure's speci�cation that is of interest,
not its implementation. \Calling" a speci�cation has no meaning. The
speci�cation of a procedure is a predicate de�ning a relation over states,
so passing p as a parameter, to a procedure R, actually means providing a
predicate representing p. This predicate can then be used (with renaming
of p's parameters) within the speci�cation of R.

In order to be able to restrict the possible values of a procedure argument,
we need to be able to talk about the speci�cation of a formal procedure
parameter in the REQUIRES clause. There is already a notation for giving the
predicate specifying a procedure, that is REQUIRES MODIFIES ENSURES, which
we can use. We extend the notion of a predicate to permit this form. This
also allows us to place restrictions on a procedure type by using a REQUIRES

MODIFIES ENSURES predicate in the type invariant.
Such a speci�cation in the REQUIRES clause represents the weakest spec-

i�cation that any actual parameter has to meet. In reasoning about client
code, the client uses the speci�cation of the actual argument in place of the
speci�cation of the formal parameter.

We allow a speci�er to refer to the components of a procedure P's
speci�cation, using P.MODIFIES, P.REQUIRES and P.ENSURES. Alternatively,
the predicate representing the full speci�cation is accessible as P.SPEC. In
either case, renaming is permitted.

As an example, consider the speci�cation of a `parameterized' sorting
routine, where the actual ordering function is given as a parameter.

We could specify this as:

<* USING TotalOrder(R for <) *>

PROCEDURE

Sort(VAR data: ARRAY OF INTEGER;

ord: PROCEDURE(a, b:INTEGER): BOOLEAN);

<* REQUIRES ord.SPEC IMPLIES

(MODIFIES NOTHING ENSURES RESULT\post = R(a, b))

AND size(data\pre) > 1

MODIFIES data

ENSURES FORALL i,j \in inds(data\pre)

ord.SPEC(data\post[i] for a,

data\post[j] for b,

i < j for RESULT\post)

AND permutation(data\pre, data\post)

*>

This speci�cation tells us that we are free to pass any function parameter
whose speci�cation is stronger than or equal to

(MODIFIES NOTHING ENSURES RESULTnpost = R(a; b))

or, in other words, has no side e�ects and implements some total order.

2.3.2 Intermediate states

In some circumstances, particularly if procedure parameters are involved
(see Section 2.3.1), it is necessary to be able to refer to states that are not
visible to the outside world, even in atomic procedures. To facilitate this,
LM3 allows predicates that are explicitly quanti�ed over states.

In such predicates, a state is bound by a quanti�er, such as s is below.
This state may be used in any place that either of the distinguished states,
pre and post, may occur. The operator, \eval, which takes a state and a
variable to a value. So, v\pre and v\post are equivalent to \eval(pre, v)

and \eval(post, v) respectively. For convenience, \eval(s, v) may be
written as v\s.

For example, if we have a speci�cation of a procedure:

PROCEDURE double(n : INTEGER) : INTEGER;

<* REQUIRES n > 0

ENSURES RESULT\post = 2 * n

*>

and a second procedure speci�ed as :

PROCEDURE

twice(p: PROCEDURE(n: INTEGER): INTEGER,

i: INTEGER): INTEGER;

<* REQUIRES p.REQUIRES(i for n) AND

FORALL s: State

(p.SPEC(s for post, i for n, j for RESULT\post)

IMPLIES p.REQUIRES(s for pre, j for n))

ENSURES EXISTS s: State

(p.SPEC(s for post, i for n)

AND p.SPEC(s for pre, RESULT\s for n)

*>

then we know that a call of twice with double as an actual value for p
is allowed, since the speci�cation of double satis�es the requirement on p.
If we wanted to perform any reasoning about this call, then p.REQUIRES, for
example, would be instantiated to the actual value of double.REQUIRES.

If we expand the predicates in such a call, we see for:

twice(double; 3)

the expanded predicate is:

9s : State
(3 > 0)

RESULTns = 2 � 3 ^
RESULTns > 0) RESULTnpost = 2 � RESULTns

which indeed simpli�es to RESULT\post = 12.
This example shows the need for quanti�cation over states since we would

be unable to refer to the intermediate result without such a mechanism.

2.3.3 Object types and methods

One of the ways in which Modula-3 di�ers from its predecessors in the
Modula family is that it has subtyping with inheritance. LM3 has to support

these features, which have not been addressed in previous Larch interface
languages.

There is a problem. There are two distinct uses of inheritance within the
Object Oriented community, only one of which represents true subtyping.
It is reasonable to assume that if a type T1 is a subtype of T, then any
properties we specify of T would still be required of T1. Unfortunately,
Modula-3 can not enforce this semantic restriction, and it is often the case
that programmers use inheritance simply to avoid rewriting some code,
without really preserving subtyping. This would make it impossible to
specify anything meaningful about the subtype relationship. LM3 supports
only disciplined use of inheritance. Anything that is speci�ed about a type
must also be true for all subtypes (modulo appropriate rebinding of rede�ned
operators). This will in certain cases require a programmer to perform
actions (such as providing a new default method) that are not required by
the programming language, per se, and certainly restricts the programmer's
freedom to override methods arbitrarily.

An object value can be regarded as a record with an associated suite
of procedures, called methods, giving operations bound to that record. An
object type has a supertype and inherits both the structure and the default
operations of this supertype. The LM3 keyword, SELF, refers to the current
instantiation of the object.

The INITIALLY clause of an object type is a post-condition on any use of
the function NEW with this type. Since this is not syntactically checkable, it
is a proof obligation. The invariant on the type must be true on instance
creation, and preserved by methods.

The speci�cation of a subtype inherits the speci�cations of its supertype
with its default methods and extends these speci�cations with specializa-
tions. Following the pattern of the Larch Shared Language, this inheritance
is treated syntactically. Semantic interpretation is on the fully expanded
form.

The grammar for an object type declaration in its simplest form, ignoring
brands, traces, etc. (see Appendix B for the full detail) is:

objectTypeDecl ::= ancestor simpleObjectType

ancestor ::= typeName j : : :

simpleObjectType ::= OBJECT �elds [methodDecl] END

methodDecl ::= METHOD method+

method ::= explicitMethod j strengthenMethodSpec

explicitMethod ::= ident signature [defaultProc] ; [procSpec]

strengthenMethodSpec ::= <� STRENGTHEN ident lm3Predicate �>

This is most easily understood in terms of an example. We specify a
somewhat arti�cial object called SetBag, which is the common ancestor of
both Set and Bag. The traits de�ning the sorts and operators follow the
interfaces.

INTERFACE SetBag;

<* USING SetBagTrait *>

IMPORT E FROM Element;

TYPE Space <: ROOT;

T <* BASED ON t:SB

INVARIANT nonNil(t) *>

= Space OBJECT

METHODS

insert(e: E) := insertDefault;

<* MODIFIES SELF

ENSURES SELF\post = add(SELF\pre, e)

*>

in(e: E): BOOLEAN := inDefault;

<* ENSURES RESULT\post = e \in SELF\pre *>

END;

PROCEDURE insertDefault(s: T; e: E);

<* MODIFIES s

ENSURES s\post = add(s\pre, e)

*>

PROCEDURE inDefault(s: T; e: E): BOOLEAN;

<* ENSURES RESULT\post = e \in s\pre *>

END SetBag.

A minimal trait is given below. More realistic traits for Set and Bag are
given in the LSL Handbook.

SetBagTrait: trait
introduces

fg : ! SB
add: SB, Elem ! SB
__\in__ : SB, Elem ! Bool

asserts

SB generated by (fg, add)
(8 s: SB, e, e1 : Elem)
:(e \in fg),
e \in add(s, e1) == (e = e1) j (e \in s)

We can then specify Set as a subtype:

INTERFACE Set;

<* USING SetTrait

IMPORT SetBag;

TYPE T <* BASED ON S *>

= SetBag.T OBJECT

METHODS

insert := setInsert;

END;

PROCEDURE setInsert(s: T; e: Elem)

<* MODIFIES s

ENSURES s\post = add(s\pre, e);

END Set.

This interface uses the trait, SetTrait, which is:

SetTrait : trait
includes SetBagTrait(S for SB)
asserts S partitioned by (\in)

Note: the post condition of the insert procedure looks the same as in
SetBag, but has a di�erent meaning since Set.T is bound to a di�erent sort
from a di�erent trait. Even if there were no explicit rede�nition, we would
still need to reinterpret the inherited predicates to ensure correct bindings
of the operators.

For this to be a valid speci�cation of a subtype, the following implications
need to hold:

1. SetTrait ) SetBagTrait

2. setInsert.SPEC ) SetBag.insert.SPEC

or, in other words, we must prove that Set is indeed a true subtype of SetBag.
We could de�ne a second subtype of SetBag, namely Bag, which adds

extra functionality.

INTERFACE Bag;

<* USING BagTrait *>

IMPORT SetBag;

TYPE T <* BASED ON B *>

= SetBag.T OBJECT

METHODS

count(e: Elem):CARDINAL := countDefault;

<* ENSURES RESULT\post =

count(SELF\pre, e) *>

END;

PROCEDURE countDefault(b: T; e: Elem): CARDINAL;

<* ENSURES RESULT\post = count(b\pre, e) *>

END Bag.

The trait for this interface is:

BagTrait : trait
includes SetBagTrait(B for SB)
introduces count : B, Elem ! Card
asserts B partitioned by count

8 (b: B, e,e1 : Elem)
count(fg, e) == 0
count(add(b,e), e1) ==

count(b,e1) + ( if e = e1 then 1 else 0)
implies 8(b : B, e,e1 : Elem)

count(b, e) > 0 )e \in b

This completes the description of the constructs of LM3. A succinct
summary of the syntax is given in Chapter 3.

Chapter 3

The Syntax of LM3

This section gives a complete description of the syntax of the LM3 language.
Whilst much of this is paraphrased directly from the Modula-3 Report, it
should be possible to follow the de�nition without knowledge of the grammar
given there. The LM3 grammar is a superset of the Modula-3 interface
grammar.

The grammar presented here is a collection of the components presented
previously; it does not go down to the token level. A complete parsing
grammar is given in Appendix B.

3.1 The LM3 reference grammar

interface ::= INTERFACE ident ; [traitUse] imports�

[intConstraints] declaration� END ident .

traitUse ::= <� USING traitRef,+ �>

imports ::= [FROM ident] IMPORT ident,+ ;

intConstraints ::= <� initial �> j <� invar �> j <� initial invar �>

declaration ::= constDecl j varDecl j typeDecl j exceptionDecl
j procDecl j privateVarDecl j : : :

initial ::= INITIALLY lm3Predicate

invar ::= INVARIANT lm3Predicate

constDecl ::= ident [: type] = constExpr

varDecl ::= VAR vd+

vd ::= ident,+ : type [initialVal] ;

initialVal ::= := expr

privateVarDecl ::= <� PRIVATE ( ident,+ : sort; varSpec)+ �>

varSpec ::= [initial] [invar]

typeDecl ::= TYPE td+

td ::= ident [typeSpec] subtypeReln type ;

typeSpec ::= <� BASED ON [ident:]sort [initialPred] [invariantPred] �>

subtypeReln ::= = j <:

type ::= ident j arrayType j recordType j : : :

procDecl ::= PROCEDURE ident ([signature]) [: type] [raisesList] ;
[<� procSpec �>]

signature ::= f[paramType] ident,+ : type [initialVal] ;g+

paramType ::= VALUE j VAR j READONLY

procSpec ::= [globals] [privates] [letDecl]

[prePred] [modi�es] postPred

postPred ::= atomicPostPred

globals ::= ((WR j RD) ident : type;)+

privates ::= PRIVATE varDecl+

letDecl ::= LET let IN

let ::= ident BE term,+

prePred ::= REQUIRES lm3Predicate

modi�es ::= MODIFIES term,+

raisesList ::= RAISES f ident,� g

atomicPostPred ::= ENSURES lm3Predicate [except] [unless]

except ::= EXCEPT fguardPredicate ) exceptionPredicateg |+

unless ::= UNLESS exceptionPredicate |+

postPred ::= atomicPostPred j compositePostPred

compositePostPred ::= COMPOSITION OF acts ;+ action+ END

acts ::= (ident[� j +];)j( acts )

action ::= ACTION ident [when] atomicPostPred

when ::= WHEN lm3Predicate

objectTypeDecl ::= ancestor simpleObjectType

ancestor ::= typeName j : : :

simpleObjectType ::= OBJECT �elds [methodDecl] END

methodDecl ::= METHOD method+

method ::= explicitMethod j strengthenMethodSpec

explicitMethod ::= ident signature [defaultProc] ; [procSpec]

strengthenMethodSpec ::= <� STRENGTHEN ident lm3Predicate �>

Chapter 4

The semantics and checking

of LM3

Thus far, most of the language has been described syntactically. In this
chapter, we give the semantics underlying the syntax. The meaning of an
interface speci�cation is given by a translation into terms in the Larch Shared
Language.

This translation, which is generated by the LM3 Checker, allows speci-
�cations to be mechanically type checked (or more accurately, sort checked
for the sorts on which the types are based). Further checking, of the kind
that requires more sophisticated tools (such as LP, the Larch Prover[5]) uses
the same mechanism but is not addressed here.

This chapter presents a set of functions that translate the LM3 text
into LSL, in a form that can be used for sort checking. We also give the
LM3 traits. These are the traits associated with the primitive types and
constructors of Modula-3.

4.1 The translation functions

This section presents the associated LSL declarations and terms for each
construct in an LM3 interface speci�cation. Each construct in the inter-
face speci�cation causes a corresponding phrase to be created in the trait,
according to the following table. This presentation is not fully formal but
indicates the translations performed by the LM3 checker.

For an interface Foo, by convention, we generate the trait in a �le called
FooTrait.lsl. Further, each trait implicitly imports LM3Trait.

For the components of an LM3 speci�cation:

INTERFACE Foo gives FooTrait: trait

USING traitList gives includes traitList

IMPORTS impList gives includes bazTrait

for each baz in impList
CONST x: T gives introduces x: ! S,

where T is based on S
VAR x: T gives introduces x: ! S,

where T is based on S
EXCEPTION e gives introduces e: ! Except

TYPE t BASED ON S gives __\pre,__\post: SRef ! S,
(REF type) __\pre,__\post: SRefVar ! SRef,

modi�es, unchanged, fresh:
SRef ! Bool,

modi�es, unchanged: SRefVar ! Bool
TYPE t BASED ON S gives __\pre,__\post: SVar ! S,
(VAL type) modi�es, unchanged: SVar ! Bool
TYPE t (M3 primitive) gives (see Section 4.2.1)
TYPE t (M3 constructor) gives (see Section 4.2.1)
INVARIANT predicate1 gives predicate instantiated in the pre state,

predicate instantiated in the post state
PROCEDURE p gives constant of the appropriate sort

(for each formal in the parameter list,
plus RESULT and RAISE)

REQUIRES predicate gives predicate as a term with each
variable fully quali�ed with its sort

MODIFIES compList gives modi�es(i: Sort) for each i in compList
ENSURES predicate gives a fully quali�ed term

(according to the expansion of the
ENSURES term given in Chapter 2)

COMPOSITION ai : : : gives isAction(ai)
ACTION a gives a:! Action, expand body as for procedure

To illustrate the translation, the Stack example is translated to the
following which then sort checks correctly using the LSL checker:

% A simple example of an interface and its translation

1Remember that this translation is for sort checking only. For full semantic checking,
the invariant predicate would be conjoined to the pre- and post-conditions of any procedure
whose formals or globals contain variables of this type.

% The example is a little perverse since the procedures insist

% on returning new stacks rather than modifying the ones they have.

% This is the way the interface is specified in the report and it's

% a good example here since the extra level of indirection forces

% one to be more careful! In this example, repeated declarations

% (e.g. s: -> RealStackRefVar) are given only once.

% `%' lines give the LM3 specification. The lines following are

% the LSL translation.

% INTERFACE Stack;

StackTrait : trait

% USING Stack(REAL for E, RealStack for C)

includes Stack(Real for E, RealStack for C)

introduces

% TYPE T <* BASED ON RealStack *>

% <: REFANY;

modifies, unchanged, fresh: RealStackRef -> Bool %since T <: REFANY

modifies, unchanged: RealStackRefVar -> Bool %for VAR parameters

__\pre, __\post: RealStackRef -> RealStack

__\pre, __\post: RealStackRefVar -> RealStackRef

% PROCEDURE Pop(VAR s:T): REAL;

s: -> RealStackRefVar

RESULT: -> Real

% PROCEDURE Push(VAR s:T; x: REAL);

x: -> Real

% PROCEDURE Create():T;

RESULT: -> RealStackRef

asserts equations

% PROCEDURE Pop(VAR s:T): REAL;

% REQUIRES NOT(isEmpty(s\pre\pre))

~isEmpty(s:RealStackRefVar\pre\pre) \;

% MODIFIES s

modifies(s: RealStackRefVar) \;

% ENSURES s\post\post = pop(s\pre\pre) AND FRESH(s\post)

% AND RESULT = top(s\pre\pre)

(s:RealStackRefVar\post\post) = pop(s:RealStackRefVar\pre\pre) &

fresh(s:RealStackRefVar\post) &

RESULT: Real = top(s:RealStackRefVar\pre\pre) \;

% PROCEDURE Push(VAR s:T; x: REAL);

% MODIFIES s

modifies(s: RealStackRefVar) \;

% ENSURES s\post\post = push(s\pre\pre,x)

% AND FRESH(s\post)

s:RealStackRefVar\post\post =

push(s:RealStackRefVar\pre\pre, x: Real) &

fresh(s:RealStackRefVar\post) \;

%

% PROCEDURE Create():T;

% ENSURES RESULT\post = new AND FRESH(RESULT\post)

RESULT:RealStackRef\post = new: RealStack &

fresh(RESULT:RealStackRef)

% END Stack.

4.2 The LM3Trait

For any user-de�ned types, the corresponding sort is given in the type spec-
i�cation in the interface. There is no actual interface giving the de�nitions
for the built-in types of Modula-3. The correspondence of such types with
sorts is actually built into the LM3 Checker, but for pedagogical reasons, we
present the speci�cations that we would associate with the built-in types if
such an interface existed.

4.2.1 The interfaces

For simple types:

TYPE

INTEGER <* BASED ON i: Int

INVARIANT MinInt <= i <= MaxInt *>;

CARDINAL <* BASED ON c: Int

INVARIANT 0 <= c <= MaxInt *>;

BOOLEAN <* BASED ON b: Bool *>;

CHAR <* BASED ON c: Char *>;

REAL <* BASED ON r: Float *>;

REFANY <* BASED ON r: RefAny *>;

TEXT <* BASED ON t: Text

INITIALLY t = empty *>;

MUTEX <* BASED ON m: Mu

INITIALLY holder(m) = none *>;

THREAD <* BASED ON Th *>;

For type constructors, the notional interface is actually a schema, since
each use of a constructor needs an appropriately instantiated sort. In some
cases, we can use the LSL shorthands to produce an appropriate trait.

In the following, we generate traits according to the indicated instanti-
ation.

ARRAY OF X <* BASED ON Array(X, XArray) *>;

SET OF X <* BASED ON Set(X, XSet) *>;

REF X <* BASED ON RefSort(X, XRef) *>;

RECORD ... <* BASED ON tuple of ... *>;

any enumeration {...} <* BASED ON enumeration of ... *>;

We made a deliberate choice not to provide a default sort for the object
constructor. Since objects are by their very nature a representation of an
abstraction, the speci�er will always provide the trait that represents that
abstraction.

Procedure types also have a special interpretation. For any procedure pa-
rameter, we generate a trait which provides the operators .SPEC, .REQUIRES,
.MODIFIES and .ENSURES which deliver Boolean results. These operators are
placeholders and are substituted by the predicates of the speci�cation of the
actual parameter in any reasoning about a use of the interface. We adopt a
convention by which the use of p.REQUIRES represents
p.REQUIRES(pre, post, p1; : : : ; pn; pRESULT). This gives default values for the
parameters and allows renaming using the mechanism from LSL. The trait
is generated by the checker, following a syntactic template.

For example, for p : PROCEDURE(INTEGER):INTEGER, we generate a trait of
the form:

ProcedureIntegerToInteger : trait
introduces

:SPEC; :REQUIRES; :MODIFIES; :ENSURES

: PIntToInt; st; st; Int; Int ! Bool

asserts

8p : PIntToInt; pre; post : State; i; j : Int

p:SPEC(pre; post; i; j) ==
(p:REQUIRES(pre; post; i; j))

p:MODIFIES(pre; post; i; j)^
p:ENSURES(pre; post; i; j))

4.2.2 The trait

LM3Trait is constructed by including the base traits, plus an instantiation
of a template for each use of a constructor.

LM3Trait : trait
includes Integer ;Boolean;Character ;Float;Text;Thread

plus instantiations of Set ;Ref ;Array;Enumeration;Procedure for

each use in an interface

The included traits are given in Appendix A.

Chapter 5

Examples

In this chapter, we present two examples of LM3 speci�cations. The �rst
presents the speci�cation of Threads, which is one of the required interfaces
for the Modula-3 implementation. Being a low-level interface this is some-
what atypical but illustrates the speci�cation of non-atomic procedures. The
second example presents a complete interface providing the functionality of
I/O streams.

5.1 The Threads interface

This example was originally presented by Birrell et al. in [1] together with an
accompanying discussion on programming with Threads. This section is a
translation of that paper to the current LM3 and much of the text and all of
the credit is due to the original authors. We present the formal speci�cation
without much commentary. This speci�cation is self-contained; none of the
informal description of threads is needed to understand its precise semantics.
However, it is intended to be used in conjunction with informal material,
such as that in [2]. The informal material provides intuition and says how
the primitives are intended to be used.

The traits used by this interface can be found in Appendix A.

5.1.1 Mutex, Acquire, Release

<* USES Mutex *>

TYPE Mutex

<* Mutex BASED ON m: Mu INITIALLY holder(m) = none *>

<: ROOT;

PROCEDURE Acquire(VAR m: Mutex);

<* MODIFIES m

WHEN holder(m\pre) = none

ENSURES holder(m\post) = CURRENT *>

PROCEDURE Release(VAR m: Mutex);

<* REQUIRES holder(m\pre) = CURRENT

MODIFIES m

ENSURES holder(m\post) = none *>

If Release(m) is executed when there are several threads waiting to
perform Acquire(m), the WHEN clause of each of them will be satis�ed. Only
one thread will hold m next, because|by atomicity of Acquire|it must
appear that one of the Acquires is executed �rst; its ENSURES clause falsi�es
the WHEN clauses of all the others. Our speci�cation does not say which of
the blocked threads will be unblocked �rst, nor when this will happen.

5.1.2 Semaphore, P, V

<* USES Semaphore *>

TYPE Semaphore

<* BASED ON s: Semaphore

INITIALLY s = unlocked *>

= { unlocked, locked };

PROCEDURE P(VAR s: Semaphore);

<* MODIFIES s

WHEN s\pre = unlocked

ENSURES s\post = locked *>

PROCEDURE V(VAR s: Semaphore);

<* MODIFIES s

ENSURES s\post = unlocked *>

5.1.3 Blocking and unblocking on condition variables

<* USES Thread, Set(Th, ThreadSet) *>

TYPE Condition

<* BASED ON c: ThreadSet

INITIALLY c = {} *>

<: ROOT;

PROCEDURE Wait(VAR m: Mutex; VAR c: Condition);

<* REQUIRES holder(m) = CURRENT

MODIFIES m, c

COMPOSITION OF Enqueue; Resume END

ACTION Enqueue

ENSURES holder(m\post) = none

AND c\post = c \union {CURRENT}

ACTION Resume

WHEN holder(m\pre) = none AND NOT(CURRENT \in c\pre)

ENSURES holder(m\post) = CURRENT AND UNCHANGED(c) *>

PROCEDURE Signal(VAR c: Condition);

<* MODIFIES c

ENSURES c\post = {} OR c\post \subset c *>

PROCEDURE Broadcast(VAR c: Condition);

<* MODIFIES c

ENSURES c\post = {} *>

Any implementation that satis�es Broadcast's speci�cation also satis�es
Signal's. We cannot strengthen Signal's postcondition: the recommended
implementation of Signal usually unblocks just one waiting thread, but may
unblock more.

5.1.4 Alerts

<* USES Thread *>

EXCEPTION Alerted;

PROCEDURE Alert(t: Thread);

<* MODIFIES t

ENSURES t.alertPending\post *>

PROCEDURE TestAlert(): BOOLEAN;

<* MODIFIES CURRENT

ENSURES IF RESULT\post

THEN CURRENT.alertPending\pre AND

NOT(CURRENT.alertPending\post)

ELSE UNCHANGED(CURRENT) *>

PROCEDURE AlertP(VAR s: Semaphore) RAISES {Alerted};

<* MODIFIES s, CURRENT

WHEN s\pre = unlocked OR CURRENT.alertPending\pre

ENSURES s\post = locked AND

UNCHANGED(CURRENT.alertPending)

UNLESS RAISE = Alerted AND CURRENT.alertPending\pre

AND NOT(CURRENT.alertPending\post) AND UNCHANGED(s) *>

The UNLESS clause in AlertP allows non-determinism.

PROCEDURE AlertWait(VAR m: Mutex; VAR c: Condition)

RAISES {Alerted};

<* REQUIRES holder(m\pre) = CURRENT

MODIFIES m, c, CURRENT

PRIVATE alertChosen: BOOLEAN

COMPOSITION OF Enqueue; ChooseOutcome; GetMutex END

ACTION Enqueue

ENSURES holder(m\post) = none AND c\post = c \union

{CURRENT}

AND UNCHANGED(CURRENT)

ACTION ChooseOutcome

WHEN NOT(CURRENT \in c\pre) OR CURRENT.alertPending\pre

ENSURES alertChosen\post = CURRENT \in c\pre

AND UNCHANGED(holder(m))

AND c\post = delete(CURRENT, c\pre)

AND CURRENT.alertPending\post = (CURRENT.alertPending\post

AND NOT(alertChosen))

ACTION GetMutex

WHEN holder(m\pre) = none

ENSURES NOT(alertChosen\pre) AND holder(m\post) = CURRENT

AND UNCHANGED(CURRENT)

UNLESS RAISE = Alerted AND alertChosen\pre

AND holder(m\post) = CURRENT) AND UNCHANGED(c)

AND UNCHANGED(CURRENT) *>

5.2 The IO Streams interface

Finally, we present a de�nition of the IO Stream interface that forms part of
the standard IO package used at the Systems Research Center. The interface
is taken from Brown & Nelson [3].

5.2.1 An IOStreams package

The package makes use of the partially opaque types of Modula-3 to present
a safe and e�cient IO package. The report describes a number of types
ranging from the abstract readers and writers, down to machine dependent
unsafe modules that exploit low-level features to achieve e�ciency. The
reader is referred to [3], both for further detail and for a good example of
well structured Modula-3 programing.

We address the input classes and present the most abstract reader and a
more concrete realization of it. We borrow enough of explanation from [3]
to make the interface comprehensible.

5.2.2 The Rd interface

Rd.T, pronounced reader, is a type that provides functions for accessing a
character input stream. Abstractly, it consists of:

len the number of source characters

src a sequence of characters

cur an integer index into src, representing the current position

avail an integer representing the number of characters available

closed a boolean that's true for a Rd that has been closed

seekable a boolean that's true if the current position can be set to anywhere
in src

intermittent a boolean that's true if the source is available in increments
rather than all at once.

Since there are many concrete representations of readers that may fail
in any number of di�erent ways, the abstract class declares an exception
Failure which takes a REFANY and is used to represent all failures.

The following is the abstract Rd interface, with some uninteresting func-
tions omitted.

Rd.i3

INTERFACE Rd; <* USING Reader *>

FROM Thread IMPORT Alerted;

TYPE T <* BASED ON rd: R

INVARIANT

intermittent(rd) OR avail(rd) = len(rd) + 1 *>

<: ROOT;

Code = {Closed, Unseekable, Intermittent, CantUnget};

EXCEPTION EndOfFile;

Failure(REFANY);

Error(Code);

PROCEDURE GetChar(rd: T): CHAR

RAISES {EndOfFile, Failure, Alerted, Error};

(* Return the next character from the src of rd *)

<* MODIFIES rd

WHEN avail(rd\pre) > cur(rd\pre)

ENSURES RESULT\post = currentVal(rd\pre)

AND rd\post = setCur(rd\pre, cur(rd\pre)+1)

AND canUnget(rd\post)

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

AND UNCHANGED(rd)

| cur(rd\pre)= len(rd\pre) => RAISE = EndOfFile

AND UNCHANGED(rd)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

PROCEDURE EOF(rd: T): BOOLEAN RAISES {Failure,Alerted,Error};

(* Return true iff rd is at end-of-file *)

<* WHEN avail(rd\pre) > cur(rd\pre)

ENSURES RESULT\post = (cur(rd\pre) = len(rd\pre))

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

PROCEDURE UnGetChar(rd: T) RAISES {Error};

(* Push back the last char read,

so the next call to GetChar will read it again *)

<* REQUIRES cur(rd\pre) > 0

MODIFIES rd

ENSURES rd\post

= setCanUnget(setCur(rd\pre, cur(rd\pre)-1), false)

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

AND UNCHANGED(rd)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Error(CantUnget)

AND NOT(canUnget(rd\pre)) *>

PROCEDURE CharsReady(rd: T): CARDINAL RAISES {Failure, Error};

(* Return some number of chars that can be read

without indefinite waiting *)

<* ENSURES IF avail(rd\pre) = cur(rd\pre)

THEN RESULT\post = 0

ELSE (1 \leq RESULT\post

AND RESULT\post

\leq (avail(rd\pre) - cur(rd\pre)))

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x) *>

PROCEDURE GetText(rd: T; len: INTEGER): TEXT

RAISES {Failure, Alerted, Error};

(* Get chars from rd until exhausted or len chars have been read *)

<* MODIFIES rd

LET inc = MIN(len, len(rd\pre) - cur(rd\pre)) IN

ENSURES (RESULT\post =

FromStr(subSrc(rd\pre, cur(rd\post))))

AND (rd\post

= setCanUnget(

setCur(rd\pre, cur(rd\pre)+ inc),

inc > 0))

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

PROCEDURE GetLine(rd: T): TEXT

RAISES {EndOfFile, Failure, Alerted, Error};

(* Read chars until newline or rd is exhausted *)

<* MODIFIES rd

ENSURES RESULT\post = FromStr(subSrc(rd\pre, cur(rd\post))

AND ((cur(rd\post) = len(rd\pre))

OR isLine(RESULT\post & NewLine))

AND rd\post =

setCanUnget(setCur(rd\pre, cur(rd\post)), true)

EXCEPT cur(rd\pre)= len(rd\pre) => RAISE = EndOfFile

| closed(rd\pre) => RAISE = Error(Closed)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

PROCEDURE GetIndex(rd: T): CARDINAL RAISES {Error};

(* Return the current index *)

<* ENSURES RESULT\post = cur(rd\pre)

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

| isNil(rd\pre) => CHECKEDRTE *>

PROCEDURE GetLength(rd: T): CARDINAL

RAISES {Failure, Alerted, Error};

(* Return the length of the src *)

<* ENSURES RESULT\post = len(rd\pre)

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

| intermittent(rd\pre) => RAISE = Error(Intermittent)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

PROCEDURE Seek(rd: T; n: CARDINAL)

RAISES {Failure, Alerted, Error};

(* Set cur to n *)

<* MODIFIES rd

ENSURES rd\post =

setCanUnget(

setCur(rd\pre, MIN(n, len(rd\pre)), false)

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

AND UNCHANGED(rd)

| NOT(seekable(rd\pre)) => RAISE = Error(Unseekable)

AND UNCHANGED(rd)

| isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

PROCEDURE Close(rd: T) RAISES {Failure, Alerted};

(* Close rd *)

<* MODIFIES rd

ENSURES rd\post = close(rd\pre)

EXCEPT isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x) AND closed(rd\post)

| RAISE = Alerted AND closed(rd\post) *>

PROCEDURE Intermittent(rd: T): BOOLEAN RAISES {};

(* Return true if rd is intermittent *)

<* ENSURES RESULT\post = intermittent(rd\pre)

EXCEPT isNil(rd\pre) => CHECKEDRTE *>

PROCEDURE Seekable(rd: T): BOOLEAN RAISES {};

(* Return true if rd is seekable *)

<* ENSURES RESULT\post = seekable(rd\pre)

EXCEPT isNil(rd\pre) => CHECKEDRTE *>

PROCEDURE Closed(rd: T): BOOLEAN RAISES {};

(* Return true if rd is closed *)

<* ENSURES RESULT\post = isClosed(close(rd\pre))

EXCEPT isNil(rd\pre) => CHECKEDRTE *>

END Rd.

Reader.lsl

The Rd interface is based on a trait that de�nes the basic operations on an
Rd.T.

Reader : trait
includes Char ;Natural; Sequence(Nat ;Char;CharSeq;Nat for Card);

Text(CharSeq); Integer(Nat for Int)
RT tuple of src : CharSeq ; cur : Nat ; avail : Nat ; closed : Bool ;

seekable : Bool ; intermittent : Bool ; canUnget : Bool
introduces

new :! R

appendSrc : R;CharSeq ! R

close : R! R

setAvail : R;Nat ! R

setCanUnget : R;Bool ! R

setCur : R;Nat ! R

avail : R! Nat

canUnget : R! Bool

closed : R! Bool

cur : R! Nat

currentVal : R! Char

intermittent : R! Bool

isNil : R! Bool

len : R! Nat

seekable : R! Bool

src : R! CharSeq

subSrc : R;Nat ! Text

proj : R! RT

asserts

8 r : R;n : Nat ; b : Bool ; cs : CharSeq
proj (appendSrc(r; cs)) == set src(proj (r); proj (r):srckcs)
proj (close(r)):closed
proj (setAvail(r; n)) == set avail(proj (r); n)
proj (setCanUnget(r; b)) == set canUnget (proj (r); b)
proj (setCur(r; n)) == set cur(proj (r); n)
avail (r) == proj (r):avail
canUnget (r) == proj (r):canUnget
closed (r) == proj (r):closed
cur (r) == proj (r):cur
currentVal(r) == (proj (r):src)[proj (r):cur ]
intermittent(r) == proj (r):intermittent
isNil(r) == r = new

len(r) == size(proj (r):src)

seekable(r) == proj (r):seekable
src(r) == proj (r):src
subSrc(r; n) ==

fromString(subsequence(proj (r):src;
proj (r):cur ;
proj (r):cur � n))

5.2.3 The RdClass interface

This interface represents a realization of the Rd abstraction, showing some
of the implementation detail. This interface illustrates the use of inheritance
and revelation within LM3 speci�cations.

RdClass.i3

RdClass reveals that every reader contains a bu�er of characters. The vari-
able buff, together with st, hi and lo represent a part of src. The invariant
describes the relationship between the representation and the abstraction
given in the supertype. Private is an opaque type that allows the hiding of
further implementation detail that is not relevant at this level.

INTERFACE RdClass <* USING ReaderClass *>;

IMPORT Rd;

FROM Thread IMPORT Alerted;

FROM Rd IMPORT Failure, Error;

TYPE Private <* BASED ON PS *>

<: ROOT;

SeekResult = {Ready, WouldBlock, Eof};

REVEAL

Rd.T <* BASED ON rd: RC

INVARIANT

FORALL i \in {lo(rd) .. hi(rd)}

(buff(rd)[st(rd) + i - lo(rd)] = src(rd)[i])

AND cur(rd) = MIN(concreteCur(rd), len(rd))

AND NOT (intermittent(rd) AND seekable(rd)) *>

= Private BRANDED OBJECT

buff: REF ARRAY OF CHAR;

st: CARDINAL; (* index into buff *)

lo, hi, cur : CARDINAL; (* indexes into src(rd)*)

closed, seekable, intermittent: BOOLEAN;

METHODS

seek(dontBlock: BOOLEAN): SeekResult

RAISES {Failure, Alerted, Error};

<* REQUIRES seekable(SELF\pre)

OR concreteCur(SELF\pre) = hi(SELF\pre)

MODIFIES SELF

ENSURES RESULT\post = Ready

AND cur(SELF\post) = concreteCur(SELF\post)

OR RESULT\post = Eof AND

cur(SELF\pre) = concreteCur(SELF\pre)

AND cur(SELF\pre) = len(SELF\pre)

OR RESULT\post = WouldBlock AND

dontBlock AND

avail(SELF\pre) = cur(SELF\pre)

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

length(): CARDINAL RAISES {Failure, Alerted, Error}

<* ENSURES RESULT\post = len(SELF\pre)

EXCEPT closed(SELF\pre) =>

RAISE = Error(Closed)

| intermittent(SELF\pre) =>

RAISE = Error(Intermittent)

| isNil(SELF\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

:= LengthDefault;

close() RAISES {Failure, Alerted, Error}

<* MODIFIES SELF

ENSURES SELF\post = close(SELF\pre)

EXCEPT isNil(SELF\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x)

AND closed(SELF\post)

| RAISE = Alerted

AND closed(SELF\post)*>

:= CloseDefault;

END;

PROCEDURE Lock(rd: Rd.T) RAISES {};

(* Lock rd *)

<* REQUIRES NOT(locked(rd\pre))

MODIFIES rd

ENSURES locked(rd\post) *>

PROCEDURE LengthDefault(rd: Rd.T): CARDINAL

RAISES {Failure, Alerted, Error}

(* A dummy default - must be overridden for any actual rd *)

<* ENSURES CHECKEDRTE

EXCEPT closed(rd\pre) => RAISE = Error(Closed)

| intermittent(rd\pre) => RAISE = Error(Intermittent)

UNLESS RAISE = Failure(x)

| RAISE = Alerted *>

PROCEDURE CloseDefault(rd: Rd.T): CARDINAL

RAISES {Failure, Alerted, Error}

(* Close rd and set the buffer to NIL *)

<* ENSURES buff(rd\post) = new AND rd\post = close(rd\pre)

EXCEPT isNil(rd\pre) => CHECKEDRTE

UNLESS RAISE = Failure(x) AND closed(rd\post)

| RAISE = Alerted AND closed(rd\post) *>

END RdClass.

ReaderClass.lsl

The trait for RdClass has much in common with Reader, adding extra
components to represent the additional �elds of Rd.

ReaderClass : trait
includes Reader ;Reader(RCforR)
RCT tuple of read : R; bu� : CharSeq ; st : Nat ; lo : Nat; hi :

Nat ; concreteCur : Nat ; locked : Bool
introduces

new :! RC

bu� : RC ! CharSeq

st : RC ! Nat

lo : RC ! Nat

hi : RC ! Nat

concreteCur : RC ! Nat

src : RC ! CharSeq

cur : RC ! Nat

len : RC ! Nat

intermittent : RC ! Bool

seekable : RC ! Bool

avail : RC ! Nat

closed : RC ! Bool

isNil : RC ! Bool

locked : RC ! Bool

lock : RC ! RC

close : RC ! RC

proj : RC ! RCT

reader : RC ! R

asserts

8 rc : RC
reader(rc) == proj (rc):read
bu� (rc) == proj (rc):bu�
st(rc) == proj (rc):st
lo(rc) == proj (rc):lo
hi (rc) == proj (rc):hi
concreteCur(rc) == proj (rc):concreteCur
locked (rc) == proj (rc):locked
locked (lock(rc))
src(rc) == src(reader(rc))
cur (rc) == cur(reader(rc))
len(rc) == len(reader(rc))
intermittent(rc) == intermittent(reader(rc))
seekable(rc) == seekable(reader(rc))
avail (rc) == avail(reader(rc))
closed (rc) == closed(reader(rc))
isNil(rc) == rc = new

proj (close(rc)):read == close(reader(rc))

Appendix A

Traits

This appendix contains most of the standard traits referenced in this report.
A few traits, such as Float, are still under development and are not ready
for inclusion here. Complete versions of these will be given in [8].

A.1 Boolean

Boolean : trait
% This trait is given for documentation only.
% It is implicit in LSL.
introduces

true; false :! Bool

: : Bool ! Bool

^ ; _ ; ) : Bool ;Bool ! Bool

asserts

Bool generated by true ; false

8 b : Bool
:true == false

:false == true

true ^ b == b

false ^ b == false

true _ b == true

false _ b == b

true ) b == b

false ) b == true

implies

AC(^;Bool);
AC(_;Bool);
Distributive(_ for +;^ for �;Bool for T );
Distributive(^ for +;_ for �;Bool for T );
Involutive(: ;Bool);
Transitive() for �;Bool for T )
8 b1; b2; b3 : Bool
:(b1 ^ b2) == :b1 _ :b2
:(b1 _ b2) == :b1 ^ :b2
b1 _ (b1 ^ b2) == b1
b1 ^ (b1 _ b2) == b1
b2 _ :b2
(b1 = b2) _ (b1 = b3) _ (b2 = b3)
b1 ) b2 == :b1 _ b2

A.2 Char

Char : trait

Ch enumeration of \000, : : : \377

A.3 Float

Float : trait

includes Integer ; DerivedOrders(R)
% The Float trait will be included in [8]

54



A.4 Integer

Integer : trait

includes TotalOrder(Int)
introduces

0; 1: ! Int

succ; pred ; � : Int ! Int

+ ; � ; � : Int; Int ! Int

asserts

Int generated by 0; succ; pred
8 x; y: Int
succ(pred(x)) == x

pred(succ(x)) == x

1 == succ(0)
x+ 0 == x

x+ succ(y) == succ(x+ y)
x+ pred(y) == pred(x+ y)
�0 == 0
�succ(x) == pred(�x)
�pred(x) == succ(�x)
x� y == x+ (�y)
x � 0 == 0
x � succ(y) == x+ (x � y)
x � pred(y) == (�x) + (x � y)
x < succ(y) == x � y

A.5 Set

Set(E;C) : trait
includes

SetBasics;

Integer;

DerivedOrders(C;� for �;� for �;� for <;� for >)
introduces

62 : E;C ! Bool

delete : E;C ! C

f g : E ! C

[ ; \ ; � : C;C! C

55



size : C ! Int

asserts

8 e; e1; e2 : E; s; s1; s2 : C
e 62 s == :(e 2 s)
feg == insert(e; fg)
e1 2 delete(e2; s) == e1 6= e2 ^ e1 2 s

e 2 (s1 [ s2) == e 2 s1 _ e 2 s2
e 2 (s1 \ s2) == e 2 s1 ^ e 2 s2
e 2 (s1 � s2) == e 2 s1 ^ e 62 s2
size(fg) == 0
size(insert(e; s)) == if e 62 s then size(s) + 1 else size(s)
s1 � s2 == s1 � s2 = fg

implies

AbelianMonoid([ for �; fg for unit ; C for T );
AC(\; C);
JoinOp([; fg for empty);
MemberOp(fg for empty);
PartialOrder(C;� for �;� for �;� for <;� for >)
C generated by fg; f g;[

8 e : E; s; s1; s2 : C
insert(e; s) 6= fg

insert(e; insert(e; s)) == insert(e; s)
s1 � s2 == s1 � s2 = fg

size(s) � 0
converts 2; 62; f g; delete; size;[;\;� : C;C ! C;�;�;�;�

A.6 Stack

Stack(E;C) : trait
includes Integer

introduces

empty :! C

push : E;C ! C

top : C ! E

pop : C ! C

len : C ! Int

isEmpty : C ! Bool

asserts

C generated by empty ; push

8 e : E; stk : C
top(push(e; stk)) == e

pop(push(e; stk)) == stk

len(empty) == 0
len(push(e; stk)) == len(stk) + 1
isEmpty(stk) == stk = empty

implies

OrderedContainer(push for insert ; top for head ; pop for tail)
C partitioned by top; pop; len

8 e : E; stk : C
len(stk) � 0
:isEmpty(push(e; stk))

converts top; pop; len

exempting top(empty); pop(empty)

A.7 Array

Array(V;VArray): trait

introduces

[ ] : VArray ; Index ! VVar

npre; npost : VArray ! VVec

[ ] : VVec; Index ! V

A.8 Ref

Ref (T;TRef ): trait

introduces

npre; npost : TRef ! T

narrow : RefAny ! TRef

widen : TRef ! RefAny

isTRef : RefAny ! Bool

Nil :! TRef

asserts

8 tr : TRef
isTRef (widen(tr))
narrow(widen(tr)) == tr

% note: for any T1 not equal T, not(isTRef(widen(t1: T1Ref)))

A.9 Text

Text(String) : trait
introduces

fromString : String ! Text

& : Text;Text ! Text

% This trait is incomplete.
% The full version will be included in [8]

A.10 Thread

ThreadTrait : trait
introduces

alertPending : Th ! Bool

A.11 Mutex

Mutex : trait
includes (Thread)
introduces

none :! Th

holder :Mu ! Th

Appendix B

A parsing grammar

This grammar is presented in a format due to Bill McKeeman (of Digital's
Technical Languages and Environments Group)[10]. It can be processed by
a tool1 into a variety of forms, such as a YACC grammar.

interface:

INTERFACE ident ; traitUse imports intCons declarations END ident .

INTERFACE ident ; traitUse intCons declarations END ident .

traitUse:

<* USING traitList *>

intCons:

initially

invariant

initially invariant

imports:

import

imports import

import:

FROM ident IMPORT idList ;

IMPORT idList ;

declarations:

1This tool is freely available. Anyone interested in a copy should send me mail at
kjones@src.dec.com.

declaration

declarations declaration

declaration:

CONST constDeclarations

TYPE typeDeclarations

EXCEPTION exceptionDeclarations

VAR variableDeclarations

procedureDeclaration

REVEAL typeDeclarations

<* specVarDeclarations *>

constDeclarations:

constDeclaration

constDeclarations constDeclaration

constDeclaration:

ident : type = constExpr ;

ident = constExpr ;

idTypeDeclaration:

: type

typeDeclarations:

typeDeclaration

typeDeclarations typeDeclaration

typeDeclaration:

ident typeSpec subTypeRelation type ;

ident subTypeRelation type ;

exceptionDeclarations:

exceptionDeclaration

exceptionDeclarations exceptionDeclaration

exceptionDeclaration:

ident ;

ident ( type ) ;

variableDeclarations:

variableDeclaration

variableDeclarations variableDeclaration

variableDeclaration:

idList : type ;

idList : type initialValue ;

specVarDeclarations:

PRIVATE specVarDeclaration

specVarDeclaration:

idList : sort varSpec

specVarDeclaration ; idList : sort varSpec

varSpec:

initially

invariant

initially invariant

initially:

INITIALLY predicate

invariant:

INVARIANT predicate

initialValue:

:= expr

subTypeRelation:

=

<:

typeSpec:

<* BASED ON sortAndVar *>

<* BASED ON sortAndVar invariant *>

<* BASED ON sortAndVar initially *>

<* BASED ON sortAndVar initially invariant *>

sortAndVar:

sort

ident : sort

procedureDeclaration:

procedureHead ;

procedureHead ; procedureSpec

procedureHead:

PROCEDURE ident signature

signature:

( )

( ) resultType

( ) raisesList

( ) resultType raisesList

( formals )

( formals ) resultType

( formals ) raisesList

( formals ) resultType raisesList

formals:

formal

formals ; formal

formal:

idList : type

idList : type initialValue

parameterType idList : type

parameterType idList : type initialValue

parameterType:

VALUE

VAR

READONLY

resultType:

: type

raisesList:

RAISES { }

RAISES { exceptionIdList }

procedureSpec:

<* globals specVarDeclarations letDeclarations prePred modifiesPred whenPred postPred *>

globals:

globals RD idList : type ;

globals WR idList : type ;

letDeclarations:

LET letDecs IN

letDecs:

ident BE expr

letDecs , ident BE expr

modifiesPred:

MODIFIES termList

MODIFIES NOTHING

MODIFIES ANY

prePred:

REQUIRES predicate

whenPred:

WHEN predicate

postPred:

atomicPost

compositePost

atomicPost:

ensuresPost

ensuresPost exceptPost

ensuresPost unlessPost

ensuresPost exceptPost unlessPost

compositePost:

compositionDefinition actionsDeclarations

ensuresPost:

ENSURES predicate

exceptPost:

EXCEPT guardedExceptionPreds

unlessPost:

UNLESS unguardedExceptionPreds

compositionDefinition:

COMPOSITION OF actionsList END

actionsList:

actionId

actionsList ; actionId

( actionsList )

actionId:

ident

ident *

actionsDeclarations:

actionDeclaration

actionsDeclarations actionDeclaration

actionDeclaration:

ACTION ident whenPred atomicPost

guardedExceptionPreds:

predicate => predicate

predicate => predicate | guardedExceptionPreds

unguardedExceptionPreds:

predicate

predicate | unguardedExceptionPreds

type:

typeName

typeName simpleObjectTypeList

UNTRACED simpleObjectTypeList

simpleObjectTypeList

arrayType

packedType

enumType

procedureType

recordType

refType

setType

subrangeType

( type )

simpleObjectTypeList:

simpleObjectType

simpleObjectTypeList simpleObjectType

arrayType:

ARRAY OF type

ARRAY [ typeIdList ] OF type

packedType:

BITS constExpr FOR type

enumType:

{ }

{ idList }

simpleObjectType:

OBJECT methodDeclarations END

OBJECT fields methodDeclarations END

brand OBJECT methodDeclarations END

brand OBJECT fields methodDeclarations END

methodDeclarations:

METHODS methodSpecs

procedureType:

PROCEDURE signature

recordType:

RECORD fields END

refType:

UNTRACED REF type

REF type

UNTRACED brand REF type

brand REF type

setType:

SET OF type

subrangeType:

[ constExpr .. constExpr ]

brand:

BRANDED

BRANDED brandName

fields:

field

fields field

field:

idList idTypeDeclaration initialValue ;

methodSpecs:

method

methodSpecs method

method:

explicitMethod

strengthenMethodSpec

explicitMethod:

ident signature defaultProc ; procedureSpec

defaultProc:

:= procedureId

strengthenMethodSpec:

<* STRENGTHEN ident predicate *>

constExpr:

expr

expr:

term

predicate:

term

termList:

term

termList , term

term:

IF term THEN term ELSE term

quantifiedTerm

logicalTerm

quantifiedTerm:

quantifier boundVarDeclarationList ( term )

quantifier:

FORALL

EXISTS

boundVarDeclarationList:

idList : type

idList : STATE

logicalTerm:

equalityTerm

logicalTerm logicalSym equalityTerm

equalityTerm:

simpleOpTerm

simpleOpTerm = simpleOpTerm

simpleOpTerm eqSym simpleOpTerm

simpleOpTerm:

prefixOpTerm

secondary postfixOps

secondary infixOpTerm

postfixOps:

simpleOp

postfixOps simpleOp

infixOpTerm:

simpleOp secondary

infixOpTerm simpleOp secondary

prefixOpTerm:

secondary

simpleOp prefixOpTerm

primary:

( term )

simpleId

simpleId ( termList )

primary selectSym simpleId

primary : sort

literal

67



secondary:

primary

bracketed

bracketed primary

primary bracketed

primary bracketed primary

bracketed:

matched : sort

matched

matched:

beginParen args endParen

beginParen endParen

beginParen:

[

openSym

endParen:

]

closeSym

args:

term

args sepSym term

args , term

simpleId:

ident

typeId:

typeName

typeName:

qualifiedId

ROOT

UNTRACED ROOT

brandName:

textLiteral

exceptionIdList:

exceptionId

exceptionId , exceptionIdList

exceptionId:

qualifiedId

procedureId:

qualifiedId

idList:

ident

idList , ident

typeIdList:

typeId

typeIdList , typeId

trait:

traitId

traitId ( renaming )

renaming:

replaceList

nameList

nameList , replaceList

nameList:

name

nameList , name

replaceList:

replace

replaceList , replace

replace:

name FOR name

name FOR name opSignature

name:

qualifiedId

opSignature:

: domain mapSym range

domain:

sortList

sortList:

sort

sortList , sort

range:

sort

traitId:

ident

traitList:

trait

traitList , trait

sort:

ident

qualifiedId:

ident

ident . ident

ident:

IDENTIFIER

literal:

TEXTLITERAL

STRINGLITERAL

NUMERICLITERAL

BOOLEANLITERAL

